Byrne Seminar, really exciting stuff! Wow, this is Experimental Math!!

Incredibly difficult homework assignment #2

And some helpful (?) suggestions

Neat and quick!!!

If you are, like me, very very lazy, you might find what follows interesting. It is, as accurately as I can type it, a *complete* `Maple` "dialog" establishing the truth of the statement $1+2+3+\ldots+n=\frac{n(n+1)}{2}$ when n is any positive integer. First I'll just give it to you uninterrupted (you can try to guess what's happening) and then I'll try to interpret each command.

> *v := n -> add(j, j = 1 .. n);*

$$v:=n\to \text{add}(j,j=1..n)$$

> *w := n->A+B*n+C*n^2;*

$$w:=n\to A+Bn+Cn^2$$

> *coefficients:=solve({seq(v(k)=w(k),k=1..3)});*

$$\text{coefficients}:=\left\{A=0,C=\frac{1}{2},B=\frac{1}{2}\right\}$$

> *w1:=n->subs(coefficients,w(n));*

$$w1:=n\to \text{subs}(\text{coefficients},w(n))$$

> *w1(n);*

$$\frac{1}{2}n+\frac{1}{2}n^2$$

> *w1(n+1)-w1(n)-(n+1);*

$$-\frac{1}{2}-n+\frac{1}{2}(n+1)^2-\frac{1}{2}n^2$$

> *expand(%);*

$$0$$

Some discussion

The first two commands, *v := n -> add(j, j = 1 .. n);* and *w := n->A+B*n+C*n^2;* define two functions just as we did in class.

The next command, *coefficients:=solve({seq(v(k)=w(k),k=1..3)});* has a huge amount of information. I'll discuss it from inside to outside.

The effect of *seq(v(k)=w(k),k=1..3)* is to get a *seq*uence of equalities: $v(1)=w(1)$, $v(2)=w(2)$, and $v(3)=w(3)$. In fact, if I just typed *seq(v(k)=w(k),k=1..3);* the result (I tried it!) would be

`1=A+B+C,3=A+2B+4C,6=A+3B+9C`

Then the *solve* command asks for a solution to these three linear equations. By the way, the punctuation inside *solve* is essential. The command *solve(seq(v(k)=w(k),k=1..3));* gets the message **Error, (in solve) invalid arguments** because apparently when *solve* gets several equations, it needs punctuation to know when the collection of equations starts and when it ends, and the { and } signal that. Remember, `Maple` is just a computer program, although, sometimes in the middle of the night, when I've been typing too much, it seems to get a personality, sometimes malicious and sometimes cute. I don't know why the order of the coefficients as they are reported is ACB. I would have guessed it would have been alphabetical. There are many mysteries! Finally, the *coefficients:=* part of the command assigns the set of solutions to the name `coefficients`.

The command *w1:=n->subs(coefficients,w(n));* creates a new function, $w1$, where the collection of coefficients has been *subs*tituted into the formula for w. I checked using the next command to see what $w1(n)$ is.
Now we need to see if $w1$ changes the way we thought it would. We already know, by the way, that $w1(1)$, $w1(2)$, and $w1(3)$ have the correct values (the *solve* command insured that). So the command *w1(n+1)-w1(n)-(n+1);* asks `Maple` to compare $w1$'s value at $n+1$ with its value at n. I subtracted $n+1$ because I hope that the result will be 0. The answer is not obviously 0, and in the final command I ask `Maple` to *expand* (sometimes the *simplify* command is useful) and the desired answer of 0 is obtained.

Advantage?

The advantage of this approach is that if we want to handle higher degrees, there will be much less typing. And I, at least, make many typing errors. There is another helpful shortcut I can show you. Look at this command and its response:

*w:=n->add(A[j]*n^j,j=0..7);*

$$w:=n \to A[0]+A[1]n+A[2]n^2+A[3]n^3+A[4]n^4+A[5]n^5+A[6]n^6+A[7]n^7$$

which is a polynomial of degree 7 with unknown coefficients*. `Maple` doesn't care – it is a program, and it will keep track of these things for you. So this is even less typing!

Difficult homework assignment

The homework assignment has seven different questions with seven different **solution teams**. Since I don't know much about you except your majors, I tried (I didn't always succeed, but tried) to create **solution teams** composed of people with different majors. Six of the solution teams have three members, and one team has two members.

Each solution team will hand in one written solution. This should contain a formula, and the `Maple` "dialog" supporting your formula. Please check my numbers in my examples. Each member of the team should sign the one written solution. Ideally, I would like the team to meet and create the solution together. If you have difficulty, please send me e-mail. I would also like some of the teams to present their solutions at the next class meeting. If your team would like to present (that's a good experience even if you are nervous) please send me e-mail. I'd like every student in the class to be a member of a team which presents a solution to a problem.

The questions and teams are on the next page. I would like you not to spend more than an hour answering your question.

I actually "know" the answer to only one of the questions but I can guess the shape of the answers to all of them. I think some of the answers could be found on the web, but I believe it's more fun to use `Maple` to guess and verify the answers.

---

* If you insist on labeling the coefficients of the polynomials alphabetically, you may run into trouble. `Maple` uses D for one of its differentiation commands. You can type *help(D);* and learn about this. If you just use the letter D alone, `Maple` may get confused!

Question #1 Use `Maple` to discover and prove a formula for the sum of the first *n odd positive integers*. For example, the sum of the first four odd integers is $1+3+5+7$, which is 16. And the sum of the first six odd integers is $1+3+5+7+9+11$, or 36.*

Solution team members here are Isita Amin and Arthur Brissac.

Question #2 Use `Maple` to discover and prove a formula for the sum of the first *n fourth powers of positive integers*. For example, the sum of the first three such integers is $1^4+2^4+3^4$, which is 98. And the sum of the first seven such integers is $1^4+2^4+3^4+4^4+5^4+6^4+7^4$, which is 4,676.*

Solution team members here are Niraj Desai, John Esmet, and Edwin Kim.

Question #3 Use `Maple` to discover and prove a formula for the sum of the first *n fifth powers of positive integers*. For example, the sum of the first four such integers is $1^5+2^5+3^5+4^5$, which is 1,300. And the sum of the first eight such integers is $1^5+2^5+3^5+4^5+5^5+6^5+7^5+8^5$, which is 61,776.*

Solution team members here are Karen Flores, Ryan Hard, and Ho Tsun Wu.

Question #4 Use `Maple` to discover and prove a formula for the sum of the first *n cubes of odd positive integers*. For example, the sum of the first three cubes of odd integers is $1^3+3^3+5^3$, which is 153. And the sum of the first six cubes of odd integers is $1^3+3^3+5^3+7^3+9^3+11^3$, or 2,556.*

Solution team members here are Nolan Campbell, Greg Zimmerman, and Alex Weiner.

Question #5 Use `Maple` to discover and prove a formula for the sum of the first *n fourth powers of integers of the form 3 plus 7 multiplied by a non-negative integer*. These integers are 3, 10, 17, 21, etc. For example, the sum of the first two of these integers is 3^4+10^4, which is 10,081. And the sum of the first five of these is $3^4+10^4+17^4+24^4+31^4$ which is 1,348,899.*

Solution team members here are Alexander Firmenich, Dipam Vyas, and Sonal Patel.

Question #6 Use `Maple` to discover and prove a formula for the sum of the first *n cubes of integers which are of the form 1 plus the square of a positive integer*. These integers are 2, 5, 10, 17, etc. For example, the sum of the first four of these integers is $2^3+5^3+10^3+17^3$, which is 6,046. And the sum of the first seven of these is $2^3+5^3+10^3+17^3+26^3+37^3+50^3$, which is 199,275.*

Solution team members here are Robert Noone, Akshata Kalia, and Ruslan Fridman.

Question #7 Use `Maple` to discover and prove a formula for the sum of the first *n squares of integers which are of the form 1 plus the cube of a positive integer*. These integers are 2, 9, 28, 65, etc. For example, the sum of the first four of these integers is $2^2+9^2+28^2+65^2$, which is 5,094. And the sum of the first seven of these is $2^3+5^3+10^3+17^3+26^3+37^3+50^3$, which is 186,395.*

Solution team members here are Parth Shah, Sollee Bae, and James Nickelson.

---

* These numbers are supplied for free by the instructor. You can use them to check the initial function defined by your `add` command. Further sums cost five cents each.